
//************************************************************
// MACRO NAME: GLOSSARY.WCM
//
// PURPOSE: To expand abbreviations.
//
// Based on code from the WPWin 5.2 distribution, but very heavily modified.
//
// Authors:
// WPCorp.
// Richard Reiner, <rreiner@nexus.yorku.ca>.
//
// Revision history:
// 5 May 93 WPC ● Released with the WPWin 5.2 distribution.
// 3 Jul 93 RjR ● Code rearrangement and cleanup for greater 
efficiency.  Abbrevs now expand in ⅓ the time!
// ● If an expansion starts with 
MacroMarkerChar, run the macro by that name (without the MacroMarkerChar) rather than 
inserting text.
//
//************************************************************

Application(WP;WPWP;Default)

//************************************************************
// Configuration section
//
// The character which, if it appears as the first char of an expansion, causes a
// macro to be run instead of text being inserted.
MacroMarkerChar:="!"

//************************************************************
// Main program
OnError(End@)
OnCancel(End@)
GetWPData(DocBlank;DocumentBlank!)
If(DocBlank)

Quit
EndIf
Call(Load@)
Call(GetAbbrev@)
Call(ExpandAbbrev@)
// Check whether the expansion starts with the macro marker
Strpos(MacroMarkerPos;MacroMarkerChar;Text)
If(MacroMarkerPos = 1)

GetWPData(MacroVariable:MacroPath;SystemVariable:MacroPath!)



EditCut()
Substr(Macroname; 2; 9999; Text)
Run(MacroPath+MacroName)

Else
Type(Text)

Endif
Call(End@)

//************************************************************
Label(GetAbbrev@)

GetWPData(SMode;SelectModeActive!)
If(not SMode)

SelectWord()
GetWPData(MacroVariable:Char; SystemVariable:LeftChar!)
If(Char = " ")

SelectCharPrevious()
Endif

Endif
GetWPData(MacroVariable:OriginalText; SystemVariable:SelectedText!)
//Convert Text to Lowercase
DllCall(SLink;"WP51STRLWR";

LowerText:STRING; {
ADDRESS(OriginalText)

}
)

Return

//************************************************************
Label(ExpandAbbrev@)

HexBuffer:=LowerText
DefText:=""
Text:=""
Call(ConvertToHex@)
DLLCall(MCRLink; 

"MCRGetIniWPString"; 
IniResult : WORD; {

"WPWP";
AnsiString("Glossary.wcm");
AnsiString(HexText);
AnsiString(DefText);
Address(Text)

}
)
If (Text = DefText)

Call(NotFound@)
Endif



Return

//************************************************************
LABEL(Load@)

DllLoad(SLink;"WPCSHAR1.DLL")
DllLoad(MCRLink;"WPCMCR.DLL")

Return

//************************************************************
Label(ConvertToHex@)

HexText:=""
ByteLen(BufferSize;HexBuffer)
DllCall(Slink;"AsciiHexFromBinary";

Result : WORD; {
Address(HexText);
Address(Hexbuffer);
Loword(BufferSize)

}
)

Return

//************************************************************
Label(NotFound@)

Title:="WordPerfect Glossary Macro"
Beep
Message:="The abbreviation """+OriginalText+""" is not defined.  Please run Glosedit.wcm to add 

this word."
Buttons:=64
DllCall(MCRLink;"MCRMessageBox";

Result:WORD; {
AnsiString(Message);
AnsiString(Title);
Buttons

}
)
Call(End@)

Return

//************************************************************
// Subr End@
Label(End@)

DllFree(SLink)
DllFree(MCRLink)
SelectMode(Off!)
Quit

Return


